

frame put — Copy selected variables or observations to a new frame

[Description](#)
[Also see](#)[Quick start](#)[Menu](#)[Syntax](#)[Remarks and examples](#)

Description

`frame put` copies a subset of variables or observations from the current frame to the specified frame. It works much like Stata's `keep` command (see [\[D\] drop](#)), except that the data in the current frame are left unchanged, while the selected variables or observations are copied to a new frame.

Quick start

Put variables `v1`, `v2`, and `v3` from the current frame into new frame `fr1`

```
frame put v1 v2 v3, into(fr1)
```

Put all variables whose name begins with `v` into new frame `fr2`

```
frame put v*, into(fr2)
```

Put all observations where `v1` is not missing into new frame `fr3`

```
frame put if !missing(v1), into(fr3)
```

Put the first observation from each cluster identified by `cvar` into new frame `fr4`

```
by cvar: frame put if _n==1, into(fr4)
```

Menu

Data > Frames Manager

Syntax

Copy selected variables from the current frame to a new frame

```
frame put varlist, into(newframename)
```

Copy observations that satisfy specified condition from the current frame to a new frame

```
frame put [varlist] if, into(newframename)
```

Copy a range of observations from the current frame to a new frame

```
frame put [varlist] in [if], into(newframename)
```

`by` is allowed with the second syntax of `frame put`; see [\[D\] by](#).

Remarks and examples

There are three main workflows for operating on a subset of data you already have in memory. One is to make use of Stata's `if` and `in` qualifiers with your commands to restrict the observations to be used. Another is to use `preserve` to make a temporary copy of the data in memory, then use `keep` and `drop` to make a subset of those data for analysis, and then to use `restore` to bring the original data back. Finally, you can leave the data in memory unchanged and use `frame put` to place a subset of the data in another frame for analysis. That frame can then be dropped, saved, or left in memory for further analysis.

`frame put` copies all variable and value labels, characteristics, and notes for any variables copied to the new frame.

► Example 1

To demonstrate `frame put`, we start with data from the 1980 U.S. Census.

```
. use https://www.stata-press.com/data/r18/census
(1980 Census data by state)
. describe
Contains data from https://www.stata-press.com/data/r18/census.dta
Observations:      50          1980 Census data by state
Variables:         13          6 Apr 2022 15:43
```

Variable name	Storage type	Display format	Value label	Variable label
state	str14	%-14s		State
state2	str2	%-2s		Two-letter state abbreviation
region	int	%-8.0g	cenreg	Census region
pop	long	%12.0gc		Population
poplt5	long	%12.0gc		Pop, < 5 year
pop5_17	long	%12.0gc		Pop, 5 to 17 years
pop18p	long	%12.0gc		Pop, 18 and older
pop65p	long	%12.0gc		Pop, 65 and older
popurban	long	%12.0gc		Urban population
medage	float	%9.2f		Median age
death	long	%12.0gc		Number of deaths
marriage	long	%12.0gc		Number of marriages
divorce	long	%12.0gc		Number of divorces

Sorted by:

We put data from several variables for all states with a population greater than 5,000,000 into new frame pop5.

```
. frame put state region pop* medage death if pop > 5000000, into(pop5)
. frame pop5: describe
Contains data
Observations:      14                1980 Census data by state
Variables:         10
```

Variable name	Storage type	Display format	Value label	Variable label
state	str14	%-14s		State
region	int	%-8.0g	cenreg	Census region
pop	long	%12.0gc		Population
poplt5	long	%12.0gc		Pop, < 5 year
pop5_17	long	%12.0gc		Pop, 5 to 17 years
pop18p	long	%12.0gc		Pop, 18 and older
pop65p	long	%12.0gc		Pop, 65 and older
popurban	long	%12.0gc		Urban population
medage	float	%9.2f		Median age
death	long	%12.0gc		Number of deaths

Sorted by:
 Note: Dataset has changed since last saved.



Also see

- [D] [frames intro](#) — Introduction to frames
- [D] [frames](#) — Data frames
- [D] [drop](#) — Drop variables or observations
- [D] [frame copy](#) — Make a copy of a frame
- [P] [frame post](#) — Post results to dataset in another frame

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).